# Maintainer : Daniel Bermond <dbermond@archlinux.org>
# Contributor: Eric Bélanger

pkgname=xine-ui
pkgver=0.99.14
pkgrel=1
pkgdesc='A free multimedia player'
arch=('loong64' 'x86_64')
license=('GPL')
url='https://www.xine-project.org'
depends=('xine-lib' 'aalib' 'curl' 'hicolor-icon-theme' 'libjpeg-turbo' 'libpng'
         'libx11' 'libxext' 'libxft' 'libxinerama' 'libxv' 'libxxf86vm' 'lirc'
         'readline')
source=("https://downloads.sourceforge.net/xine/${pkgname}-${pkgver}.tar.xz")
sha256sums=('d4d490d5cece70e2bb9849c9e482f2cf87af0302d451b614476fdcc3642cd9c3')

build() {
    cd "${pkgname}-${pkgver}"
    ./configure --prefix='/usr' --with-x --without-caca
    make
}

package() {
    make -C "${pkgname}-${pkgver}" DESTDIR="$pkgdir" install
}
